

NASA Technical Memorandum 104596

An Earth Albedo Model

A Mathematical Model for the Radiant Energy Input to an Orbiting Spacecraft Due to the Diffuse Reflectance of Solar Radiation From the Earth Below

Thomas W. Flatley and Wendy A. Moore

JANUARY 1994

(NASA-TM-104596) AN EARTH ALBEDO
MODEL: A MATHEMATICAL MODEL FOR THE
RADIANT ENERGY INPUT TO AN ORBITING
SPACECRAFT DUE TO THE DIFFUSE
REFLECTANCE OF SOLAR RADIATION FROM
THE EARTH BELOW (NASA) 33 p

N94-24497

Unclas

G3/12 0202115





NASA Technical Memorandum 104596

An Earth Albedo Model

A Mathematical Model for the Radiant Energy Input to an Orbiting Spacecraft Due to the Diffuse Reflectance of Solar Radiation From the Earth Below

**Thomas W. Flatley
Wendy A. Moore
*Goddard Space Flight Center
Greenbelt, Maryland***



**National Aeronautics and
Space Administration**

**Goddard Space Flight Center
Greenbelt, Maryland**

1994




Table of Contents

Introduction.....	1
Albedo.....	1
Diffuse Reflectance.....	2
Conditions.....	4
Sun Sensors.....	5
Calculations.....	7
Dividing the Field of View.....	7
Orbital Coordinate Frame.....	12
Calculating \hat{u}	13
Calculating \hat{n}_e	13
Implementing the Algorithm.....	14
Conclusions.....	15
Table of Variables.....	17
References.....	21
Appendix I : Operational Notes.....	23
Albedo Fortran Program.....	27
Appendix II : Reflectance Data.....	33

AN EARTH ALBEDO MODEL

Introduction

Coarse Sun Sensors are often used on spacecraft as part of Attitude Determination Systems. They function essentially as "Direction Cosine" sensors, with their output approximately proportional to the cosine of the angle between the "boresight" of the sensor and a vector from the spacecraft to the sun. Their field of view is approximately hemispherical.

The sensing element is typically a silicon solar cell which produces a current proportional to the energy flux incident on its surface. Energy can arrive however from sources other than the sun which lie within the field of view of the sensor. For an Earth-orbiting spacecraft, additional currents could be produced by such things as the Moon, sunlight reflected from some part of the spacecraft and/or radiation from portions of the Earth's surface.

This report presents a mathematical model for the Coarse Sun Sensor output due to radiation originating from the sunlit portion of the Earth within the sensor field of view.

Albedo

Solar radiation arriving at the Earth's surface is generally considered to be partially absorbed, partially specularly reflected and partially diffusely reflected. Local surface characteristics and cloud cover conditions determine the relative importance of these phenomena.


The energy which is absorbed is eventually re-radiated into space at infrared wavelengths. Solar cells are insensitive to this radiation.

With specular reflection (as commonly occurs with mirrored surfaces) some or all of the incoming solar rays are reflected with the angle of reflection equal to the angle of incidence. Since a spacecraft would receive very little energy from even an entire Earth which was specularly reflecting this type of reflection is ignored here.


Here, we consider the sunlit portion of the Earth to be a uniform, diffuse reflector and will use the word "albedo" in a limited sense, i.e. the albedo constant will be taken to be the ratio of the energy diffusely radiated from a surface to the total energy incident on the surface.

Diffuse reflectance

Diffuse reflectance is due to the scattering of the incident light in all directions. Consider a small area of the Earth's surface, dA_e . As the diffuse reflectance scatters, it forms a 2π steradian solid angle with constant intensity levels as shown. As the light travels outward, it spreads out, reducing the intensity.



To determine the strength of the diffuse reflectance at the spacecraft's position, consider the geometry shown;



where, \hat{n}_e : unit vector normal to dA_e
 \hat{s} : unit vector from the Earth to the sun
 B : point where \hat{s} intersects the Earth's surface

According to Wertz, F_{sun} , the solar constant in the vicinity of the Earth, is approximately 1358 watts/m^2 . The sunlight strikes the Earth with this intensity at point B. At locations away from this point, the intensity of the incoming sunlight decreases proportional to $\cos\psi$, so that the solar flux reaching any given incremental area is:


$$F_{\text{in}} = F_{\text{sun}}(\hat{n}_e \cdot \hat{s}) \text{ watts/m}^2$$

This incoming solar flux is partially absorbed and partially reflected. The amount of light reflected is proportional to the incident light by an albedo constant, ALB, which depends on the Earth's surface characteristics. (See Appendix II.) **This model assumes that the albedo constant does not vary over the Earth's surface, neglecting the variation of diffuse reflectance with geographical features.** A good estimate of the Earth's annual average albedo constant is 0.3.

Under these assumptions, the amount of solar flux diffusely reflected from dA_e is given by:

$$\begin{aligned} F_{\text{out}} &= \text{ALB}(F_{\text{in}}) \\ &= \text{ALB}(F_{\text{sun}})(\hat{n}_e \cdot \hat{s}) \text{ watts/m}^2 \end{aligned}$$

As the light travels outward, it spreads out, reducing its flux, so only a portion of the diffuse reflectance from dA_e actually reaches the spacecraft's position. Consider a hemisphere centered at dA_e with radius D , the distance from dA_e to the spacecraft's position.




$$\begin{aligned} \text{total energy over this hemisphere} &= \int_{\phi=0}^{\frac{\pi}{2}} \omega_0 \cos \phi (2\pi D \sin \phi) D d\phi \\ &= \pi D^2 \omega_0 \text{ watts} \end{aligned}$$

By conservation of energy, the total energy over this hemisphere must equal the energy radiated from dA_e . Using this fact, solve for ω_0 :

$$\begin{aligned} \pi D^2 \omega_0 &= F_{out} dA_e \\ \omega_0 &= \frac{ALB(F_{sun})(\hat{n}_e \cdot \hat{s}) dA_e}{\pi D^2} \text{ watts/m}^2 \end{aligned}$$

The solid angle subtended by dA_e from the spacecraft's point of view, is given by;

$$dA = \frac{dA_e \cos \Theta}{D^2}$$




So, the flux at the spacecraft's position is:

$$\begin{aligned} \omega_{alb} &= \omega_0 \cos \Theta \\ \omega_{alb} &= \frac{ALB(F_{sun})(\hat{n}_e \cdot \hat{s}) dA}{\pi} \text{ watts/m}^2 \end{aligned}$$


Conditions

The diffuse reflectance from dA_e only affects the output current of the sun sensors if the light reaches the sun sensors. For this to occur, the area, dA_e must meet the following conditions:


- 1) Assume a perfectly spherical Earth. The area, dA_e , must be on the sunlit side: $(\hat{s} \cdot \hat{n}_e) > 0$



- 2) The area must be in the spacecraft's field of view: $(\alpha < \alpha_{max})$




- 3) Assume that a Coarse Sun Sensor (CSS) has a conical field of view with a half angle of Δ . The reflected light must be in the CSS field of view: $(\hat{u} \cdot \hat{n}_{CSS}) \geq \cos \Delta$



Sun Sensors

This algorithm models Coarse Sun Sensors. In general, the Coarse Sun Sensor's output current, I_{CSS} , is proportional to $\cos\beta$, where β is the angle between the CSS boresight and \hat{u} , the direction of light source:




$$I_{CSS} \propto (\omega_{CSS}) \cos\beta$$

where, ω_{CSS} : input flux to CSS


\hat{n}_{CSS} : unit vector normal to the CSS

\hat{u} : unit vector from the CSS to the light source

Coarse sun sensors typically produce output currents (in the microampere range) when illuminated by direct and/or reflected solar energy. In this analysis, the output current (I_{CSS}) is normalized such that its value is 1.0 when the sun lies along the boresight of the CSS and there is no additional heat input from any other source.



In reality, the CSS receives incoming light from Earth as well as from the sun. Consider a small area of the Earth as the sole light source.



In this case, the incoming light is albedo rather than sunlight, so $\omega_{CSS} = \omega_{alb}$, the energy flux due to albedo at the spacecraft's position.

If sunlight along the boresight produces a current of I_{max} , the maximum CSS current due to albedo is:

$$\frac{I_{max}}{F_{sun}} = \frac{dI_{albmax}}{\omega_{alb}}$$

$$dI_{albmax} = \frac{I_{max}(ALB)(\hat{n}_e \cdot \hat{s})dA}{\pi}$$

Since, I_{CSS} is proportional to β , use the maximum values to find an appropriate proportionality constant for dI_{alb} , the CSS output current due to albedo.

$$dI_{alb} = K(\omega_{alb})\cos\beta$$

Substitute the maximum values to find K. Assume that the direction of the incoming light coincides with the CSS boresight such that $\beta = 0$:

$$\frac{(I_{max})(ALB)(\hat{n}_e \cdot \hat{s})dA}{\pi} = K \left[\frac{ALB(F_{sun})(\hat{n}_e \cdot \hat{s})dA}{\pi} \right]$$

$$K = \frac{I_{max}}{F_{sun}} \frac{\text{amperes}}{\text{watts/m}^2}$$

In general, CSS output current is calculated:


$$I_{css} = K(\omega)\cos\beta$$

Note that $\cos\beta = (\hat{n}_{css} \cdot \hat{u})$; substitution into the general equation, gives the CSS current due to albedo from dA_e as:

$$dI_{alb} = \frac{I_{max}(ALB)(\hat{n}_e \cdot \hat{s})(\hat{n}_{css} \cdot \hat{u})dA}{\pi} \text{ amperes}$$

Calculations


Consider the Earth as a light source. To calculate dI_{alb} , it is necessary to know \hat{u} , the direction of the incoming light. At great distances, the Earth may be considered a point source; however, at orbital altitudes the curvature of the Earth must be taken into account. To accomplish this, the Earth is divided into incremental areas; each reflects light at a slightly different angle. The CSS current due to albedo also depends on the strength of the diffuse reflectance, which is a function of ALB and F_{in} . As previously discussed, it is necessary to define \hat{n}_e , a unit vector normal to a given area if the Earth, to calculate the strength of the incoming solar flux.



The incremental areas are found by dividing the solid angle subtended by the Earth from the spacecraft's point of view into NA solid angles of equal size (dA), each one corresponding to a piece of the Earth (dA_e). Once the areas are defined, the necessary unit vectors are calculated for each area.

Dividing the Field of View

If the spacecraft is in a circular orbit, the size of the spacecraft's field of view remains constant, so that once these areas are defined they are fixed. If α_{max} is the angle from the subsatellite point to the horizon for a spacecraft at altitude, $ALTP$, above a spherical earth of radius R_e , then:




$$\sin \alpha_{max} = \frac{R_e}{(R_e + ALTP)}$$

From this, the solid angle subtended by the earth is calculated:

$$A_{\max} = 2\pi(1 - \cos\alpha_{\max}).$$


Consider a sphere of unit radius centered on the spacecraft, where the solid angle subtended the Earth defines a spherical segment with a surface area of A_{\max} :



To divide the field of view into NA equal parts, a set of N concentric circles ($i=1,N$) is drawn on the spherical segment, where the largest circle corresponds to the horizon of the Earth. Each circle defines another spherical sub-segment and represents the intersection of a cone with the unit sphere.

Next, each sub-segment must be subdivided into equal solid angles.

To understand how this system of concentric circles may be subdivided into NA equal areas, consider a flat circle:



Start with a circle of unit area. Add a circle three times larger. The area enclosed by the larger circle is nine units. The area between the two circles is eight units. If this latter area is divided into eight equal parts, the resultant figure contains nine equal areas.

Define the centers of these areas as; one at the center of the unit circle and eight others spaced around the dashed circle which divides each area into two equal halves; the centers are located at the points where this dashed circle intersects the radial lines which bisect each area. In this case, the dashed circle must enclose an area of 5 units. Let r_2 be the radius of the dashed circle. Thus,

$$\pi r_2^2 = 5\pi r_1^2$$

$$r_2 = r_1\sqrt{5}$$

Thus, its radius will be $\sqrt{5}$ times that of the original unit circle.

In general, let : N = total number of circles

NA = total number of equal areas

R_i = relative radius of i^{th} circle

r_i = relative radius of circle locating the centers of the i^{th} annulus areas

θ_i = angle between radial dividing lines of the i^{th} circle

Next, add a circle of relative radius $R_i = 5$. The total area of the new annulus formed will be $25 - 9 = 16$ units. Divide this annulus into 16 equal parts. Now, the figure will have 25 equal areas. The centers of these new areas will lie on a circle of radius, $r_i = \sqrt{17}$


Continuing to add circles which form equal width annuluses and dividing each into eight more sections than the previous circle, the result is:

i	R_i	r_i	θ_i
2	3	$\sqrt{5}$	45.0°
3	5	$\sqrt{17}$	22.5°
4	7	$\sqrt{37}$	15.0°
5	9	$\sqrt{65}$	11.25°
6	11	$\sqrt{101}$	9.0°

In general,


$NA = (2N-1)^2$ $r_i = \sqrt{4(i-1)^2+1}, i = 2 \text{ to } N$ $\theta_i = \frac{360}{8(i-1)}, i = 2 \text{ to } N$

Now, extend this planar analogy to the spherical segment. Begin with the spherical segment with a half cone angle of α_{\max} , corresponding to the spacecraft's field of view of the Earth. The surface area of this unit cone is equivalent to the solid angle subtended by the Earth, A_{\max} .



If this area is to be divided into NA equal parts, then the area of one unit is:

$$A_1 = \frac{A_{\max}}{NA}$$




Let α_1 define a central sub-segment of area, A_1 . Then,

$$2\pi(1 - \cos\alpha_1) = \frac{A_{\max}}{NA}$$

$$\cos\alpha_1 = \frac{NA - 1 + \cos\alpha_{\max}}{NA}$$

Define the corresponding unit vector, \hat{u}_1 , through the center of the area. In

this case, \hat{u}_1 is along the axis of the field of view cone.




Now find a cone which encloses an area of $5A_1$; this is analogous to the first dashed circle, corresponding to $i = 2$:

$$2\pi(1 - \cos\alpha_2) = \frac{5A_{\max}}{NA}$$

$$1 - \cos\alpha_2 = \frac{5}{NA}(1 - \cos\alpha_{\max})$$

$$\cos\alpha_2 = \frac{NA - 5 + 5\cos\alpha_{\max}}{NA}$$



In addition to \hat{u}_1 , define eight more unit vectors at an angle of α_2 from the axis of the field of view cone, equally spaced around the axis in increments of θ_2 , where $\theta_2 = \frac{360^\circ}{8}$.

These are the first nine unit vectors.

Next, find a cone which encloses an area of $17A_1$; following the planar example, the centers of these areas will be evenly spaced around this circle in increments of θ_3 , where $\theta_3 = \frac{360^\circ}{16}$. In this case;


$$\cos \alpha_3 = \frac{NA - 17 + 17 \cos \alpha_{\max}}{NA}$$

resulting in 16 unit vectors at an angle α_3 from the axis.

In general, there are NA solid angles of equal size. To define the appropriate unit vectors it is necessary to calculate the location of the center of each dA .

For the k^{th} solid angle ($k=1, NA$), the center's coordinates are (γ_k, δ_k) , where γ_k is the angle from the axis of the unit field of view cone to the center of the k^{th} dA and δ_k is the radial angle measured counterclockwise from the \hat{i} axis to the center of the k^{th} dA .

The angular coordinate is calculated as follows:




$$\begin{aligned} r_i &= \text{distance to center of } k^{\text{th}} \text{ area} \\ &\quad (\text{constant for each circle}) \\ &= \sqrt{4(i-1)^2 + 1} \end{aligned}$$

$\therefore \gamma_k$ coordinate of k^{th} area is:

$$\cos \gamma_k = \cos \alpha_i = \frac{NA - r_i^2 + r_i^2 (\cos \alpha_{\max})}{NA} \quad \text{for all areas in the } i^{\text{th}} \text{ circle, where } i = 2 \text{ to } N$$

To calculate the size of the radial division of the i^{th} circle, let;




$$\begin{aligned}
 M &= \text{number of radial divisions in } i^{\text{th}} \text{ circle} \\
 &= 8(i - 1) \\
 \theta_i &= \text{size of radial division in } i^{\text{th}} \text{ circle} \\
 &= \frac{360^\circ}{M}
 \end{aligned}$$

$\therefore \delta_k$ coordinate of k^{th} area is $\delta_k = j\theta_i - \frac{\theta_i}{2}$ where $j = 1$ to M for the i^{th} circle.

Orbital Coordinate Frame

The unit vectors are calculated in a 1-2-3 orbital coordinate system, which moves with the spacecraft. If the orbit is circular, the vectors are fixed within this frame, and \hat{n}_o and \hat{u} need only be calculated once, then transformed to the spacecraft body frame at each time step.

The axes of this 1-2-3 frame are defined as:



$$\begin{aligned}
 \hat{1} &= \hat{H}_o \times \hat{R} = (\hat{R} \times \hat{V}) \times \hat{R} = \hat{2} \times \hat{3} \\
 \hat{2} &= \hat{H}_o = \hat{R} \times \hat{V} \\
 \hat{3} &= \hat{R}
 \end{aligned}$$

where, \hat{R} : spacecraft unit position vector in earth centered coordinates

\hat{V} : spacecraft velocity unit vector

\hat{H}_o : unit orbit normal vector


Calculating \hat{u}

In the 1-2-3 coordinate frame, a unit vector from the spacecraft to dA_E is:

$$\hat{u}_k = \begin{bmatrix} \sin\gamma_k \cos\delta_k \\ \sin\gamma_k \sin\delta_k \\ -\cos\gamma_k \end{bmatrix}_{123 \text{ frame}}$$

Calculating \hat{n}_e

Once \hat{u} has been found for each area, calculate \hat{n}_e , the normal to each area. First, consider the 1-2-3 coordinate frame. In general, any point G in this coordinate system may be located by \bar{G}_{123} , a vector from the Earth's center to the point, G.




$$\bar{G}_{123} = \begin{bmatrix} Du_1 \\ Du_2 \\ R + Du_3 \end{bmatrix}$$

$$\bar{G}_{123} = \bar{R} + D\hat{u}$$

where D is the distance from the spacecraft to the point G.

To define the normal vector, G must be

a point on the Earth's surface



$$\therefore |\bar{G}| = R_e$$


$$\sqrt{(Du_1)^2 + (Du_2)^2 + (Du_3 + R)^2} = R_e$$

Solve this quadratic for D:

$$D = -Ru_3 \pm \sqrt{(Ru_3)^2 - (R^2 - R_e^2)}$$

Choose the smaller value of D, corresponding to G, not G'.

Once the appropriate D is calculated, the normal unit vector is:



$$\hat{n}_e = \frac{1}{R_e} \begin{bmatrix} Du_1 \\ Du_2 \\ R + Du_3 \end{bmatrix}$$

where


$$D = -Ru_3 + \sqrt{(Ru_3)^2 - (R^2 - R_e^2)}$$

Implementing the Algorithm

Appendix I contains a copy of the FORTRAN subroutine based on this paper. The table of variables provides a cross reference between the notation in the paper and the subroutine variables.

In order to optimize the calculations, the subroutine uses the conditions to determine which areas actually affect a given CSS, then calculates the output current based on only those areas.

During the first call, the subroutine initializes constants and calculates the unit vectors, \hat{u}_k and \hat{n}_{ek} , for each of the NA areas ($k=1, NA$). To calculate these vectors, begin with the first circle ($i=1$), and place \hat{u}_1 along the axis of the spacecraft's field of view cone. This corresponds to the vector $(0,0,-1)$ in the 1-2-3 orbital frame.



$$(\gamma_2, \delta_2) = \left(\alpha_2, \frac{\theta_2}{2} \right)$$

$$(\gamma_3, \delta_3) = \left(\alpha_3, \frac{3\theta_2}{2} \right)$$

For the second circle ($i=2$), calculate α_2 and θ_2 , the size of the angular and radial divisions for this circle. Then, find the coordinates (α_k, δ_k) of the centers of the M areas comprising this circle. For each set of coordinates, calculate \hat{u}_k , in the 1-2-3 orbital frame. Continue this process, until \hat{u}_k has been defined for each radial division in the second circle, (This corresponds to areas $k=2,9$). Move to the next circle and repeat this process until \hat{u}_k has been found for all areas, 1 through NA. Once all of the \hat{u}_k vectors are calculated, the corresponding

\hat{n}_{e_k} vectors are found. By defining the areas based on the spacecraft's field of view, not only is Condition 2 automatically satisfied, but the smallest possible area of the Earth's surface is considered. This ends the initialization process.

At each call, the subroutine calculates the sun unit vector in the 1-2-3 orbital frame; the position of the sun vector is necessary to test Condition 2. For the j^{th} CSS ($l=1,P$), use the remaining conditions to determine if the albedo from the k^{th} area affects this CSS. First, determine if the k^{th} area is lit (Condition 1). If so, determine if it falls within this CSS's field of view (Condition 3). Next, check that the incoming light is not blocked. (Refer to the Operational Notes in Appendix I for a complete explanation of blockage.) If all of these conditions are met, calculate dl_{alb} from this area. Once dl_{alb} has been calculated for areas 1 through NA, sum these currents to calculate I_{css} , the total CSS output current due to albedo for the j^{th} sun sensor. Repeat this process for each CSS, 1 through P.

Conclusions

This simplified albedo model was developed for use in spacecraft control system simulations, specifically, for modeling Coarse Sun Sensors. It is based on several approximations. Only diffuse reflectance is included; specular reflectance is neglected. For an elliptical orbit, the unit vectors associated with the incremental areas should change direction with altitude; instead, this algorithm assumes a circular orbit. The albedo constant is set to the annual global average for the entire Earth; Appendix II illustrates how the percentage of light reflected truly varies with geographical features. The Earth is considered a perfect sphere which does not rotate; it was unnecessary to model rotation since the albedo constant was not varied.



Table of Variables

Units are given in parentheses.

The equivalent variables from the albedo subroutine are given in brackets.

A : transformation matrix from spacecraft body coordinates to the 1-2-3 orbital frame [A]

A_{\max} : solid angle subtended by the portion of the Earth's surface, visible from the spacecraft (steradian) [AMAX]

A_1 : area of central subsegment of unit sphere, define as one unit area, equal to the size of dA (steradian) [A!]

ALB : ratio of diffuse reflectance to incident light [ALB]

ALTP : altitude of the spacecraft at perigee (km) [ALTP]

B : point where \hat{s} intersects the Earth's surface

BLOCK : a Px4 matrix where the first three elements are the components of the unit vector along the axis of the blockage cone in spacecraft body coordinates, and the fourth element is the half angle of the blockage cone in degrees [BLOCK]

BLOCKAGE : flag used to indicate whether the Coarse Sun Sensor's field of view is partially blocked [BLOCKAGE]

D : distance from the k^{th} dA_E to the spacecraft's position (km) [D]

dA : solid angle subtended by the area dA_E in the spacecraft's field of view (steradian)

dA_E : a piece of the sunlit Earth's surface (m^2)

dI_{alb} : normalized Coarse Sun Sensor output current due to the albedo from a single dA_E (amperes) [DI]

dI_{albmax} : maximum possible Coarse Sun Sensor output current due to the albedo from a single dA_E (amperes) [IMAX]

F_{in} : solar flux input to a given dA_E (watts/m^2)

F_{out} : solar flux output from a given dA_E (watts/m^2)

F_{sun} : solar constant in the vicinity of the Earth (watts/m^2)

G : any point in the 1-2-3 orbital coordinate frame

\vec{G}_{123} : a vector from the Earth's center to the point, G, in the 1-2-3 orbital frame

\hat{H}_o : unit orbit normal vector [HO]

I_{max} : maximum Coarse Sun Sensor output current. (amperes) [CSSIMAX]
 I_{css} : Coarse Sun Sensor output current (amperes) [ALBICSS]
 j : refers to the j^{th} radial area in the i^{th} annulus [J]
 K : proportionality constant for normalized Coarse Sun Sensor output current due to the Earth's albedo [KAPPA]
 k : refers to the k^{th} unit area out of NA equal areas [K]
 l : refers to the l^{th} Coarse Sun Sensor of out P Coarse Sun Sensors
 M : number of radial division in the i^{th} annulus circle [M]
 N : number of concentric circles [N]
 NA : number of equal areas Earth is divided into [NA]
 \hat{n}_{css} : unit vector normal to the Coarse Sun Sensor [NCSS]
 \hat{n}_e : outward pointing unit vector normal to dA_e [NORM]
 P : number of Coarse Sun Sensors [P]
 R_i : relative radius of i^{th} circle
 \hat{R} : unit vector from the Earth's center to the spacecraft's position [RB]
 \bar{R} : vector from the Earth's center to the spacecraft's position (km) [RMAG = $|\bar{R}|$]
 R_e : radius of the Earth (Km) [RE]
 r_i : perpendicular distance from the axis of the spacecraft field of view cone to the center of the i^{th} annulus; given in terms of r_1 [RI]
 r_1 : radius of central circle of unit area
 \hat{s} : unit vector from the Earth to the sun [SB]
 \hat{u} : in general, a unit vector from the Coarse Sun Sensor to the source of the incoming light; specifically, it is a unit vector from the spacecraft to the k^{th} dA_e [U]
 \hat{V} : spacecraft velocity unit vector [VB]
 α_i : angle from the axis of the unit radius spacecraft field of view cone to the center of the i^{th} annulus
 α_{max} : half angle of the cone encompassing the spherical segment; the angle from the nadir to the horizon [ALPHAMAX]
 α_1 : half angle of cone encompassing a central sub-segment of unit area
 β : angle between \hat{u} and \hat{n}_{css}
 Δ : half angle of Coarse Sun Sensor's conical field of view [CSSLM = $\cos\Delta$]

- δ_k : radial coordinate of the center of the k^{th} dA; the angle measured counterclockwise from the \hat{i} axis to the center of the k^{th} dA
- γ_k : angular coordinate of the center of the k^{th} dA; the angle from the axis of the unit radius spacecraft field of view cone to the center of k^{th} area; it is constant for all areas in the i^{th} circle [CGAMMA and SGAMMA, cosine and sine of γ_k]
- Θ : angle between the k^{th} \hat{n}_e and the spacecraft's position
- θ_i : size of the radial division of the i^{th} circle [THETA]
- ϕ : variable of integration
- ω_{alb} : flux at the spacecraft's position (watts/m^2)
- ω_{css} : flux input to the CSS (watts/m^2)
- ω_o : maximum flux at distance D (watts/m^2)
- ψ : angle between \hat{s} and the k^{th} \hat{n}_e




References


Earth Albedo and Emitted Radiation. NASA SP-8067. July, 1971.

Wertz, James (ed.). Spacecraft Attitude Determination and Control. D. Riedel Publishing, Co. 1980.

PRECEDING PAGE BLANK NOT FILMED



Appendix I



Operational Notes

Before running the subroutine the user **MUST** set the following constants:

BLOCKAGE : 0 = no blockage; 1 = blockage exists

N : number of circles. **DO NOT EXCEED N=16 or all arrays currently dimensioned to 1000 must be redimensioned to NA, where NA = $(2N - 1)^2$**

CSSLM : cosine of the half angle of the Coarse Sun Sensor field of view
($\cos\Delta$)

P : number of CSS. **DO NOT EXCEED 8 or all arrays currently dimensioned to 8 must be redimensioned to P.**

ALTP : Altitude of the spacecraft at orbit perigee in kilometers.


ALB : albedo constant

CSSIMAX : CSS current with full Sun directly along CSS boresight

All of these constants, except BLOCKAGE, are used are previously defined. BLOCKAGE is a switch, indicating whether the CSS's field of view is partially blocked. This subroutine models blockage as a cone, as illustrated below.

If blockage exists, the locations must be listed in the BLOCK matrix, which is passed in at each call. If the CSS are not blocked, simply set the BLOCKAGE flag to zero, and zero the BLOCK matrix.

The BLOCK matrix is a Px4 array where P is the number of CSS.



The first, second, and third elements are the components of a unit vector along the axis of the blockage cone in spacecraft body coordinates. The fourth element is λ , the half-angle of the blockage cone:

CSS	$X_{s/c}$	$Y_{s/c}$	$Z_{s/c}$	λ deg
1
2
.
P

The following variables are passed in at each call:

T : time

NCSS : CSS boresight unit vectors in spacecraft body coordinates

BLOCK : array of CSS blockage values

VB : spacecraft velocity unit vector in spacecraft body coordinates

RB : spacecraft position unit vector in spacecraft body coordinates

SB : sun unit vector in spacecraft body coordinates

The albedo subroutine returns

ALBICSS : final CSS currents due to albedo (in same units as CSSIMAX)

Albedo Subroutine

SUBROUTINE ALBEDO(T,NCSS,ALBICSS,BLOCK,VB,RB,SB)

* This subroutine approximates the current in Coarse Sun Sensors due
* to the Earth's albedo. Only includes albedo due to diffuse reflectance.
*** ASSUME:
* neglect albedo due to specular reflectance
* circular orbit
* perfectly spherical earth
* conical field of view for coarse sun sensors
*** CONSTANTS:
* ALPHAMAX : S/C field of view half angle
* AMAX : solid angle subtended by the portion of the Earth's
* surface visible from the spacecraft (steradian)
* DA : solid angle subtended by dAe, a small piece of sunlit
* earth, from the spacecraft's field of view (steradian)
* RE : radius of the earth (km)
* RMAG : distance from center of earth to S/C in circular orbit
** User sets these constants in the ALBEDO subroutine:
* ALB : albedo constant; ratio of diffuse reflectance to incident
* light
* ALTP : altitude of spacecraft at perigee (km)
* BLOCKAGE: flag to indicate whether the CSS are blocked. [0.0 = no,
* 1.0 = yes] If they are the blockage numbers must be in
* the (P,4) array BLOCK
* CSSIMAX : maximum current output by CSS to Attitude Control System
* when sun lies along CSS boresight. (Include any scaling by
* Attitude Control Electronics.)
* CSSLM : cosine of the half-angle of the CSS FOV; ALBEDO assumes
* all of the P CSS have the same size conical FOV
* N : number of concentric circles
* NA : number of equal areas
* P : number of coarse sun sensors
** User passes in these constants from the calling routine at each time
** step:
* NCSS : unit vectors of CSS boresight in S/C body coordinates
*** VARIABLES:
** User passes in these variables at each time step:
* BLOCK : a Px4 array which holds the blockage values for the CSS
* Modeled as a blockage 'cone'. First three elements of each
* row are the S/C body coordinates of this cone's axis.
* The fourth elements is the angle of the blockage cone
* in degrees.
* RB : spacecraft position unit vector from earth center to S/C in
* S/C body coordinates
* SB : unit vector from the Earth to the sun in S/C body coordinates

* T : time
 * VB : spacecraft velocity unit vector in S/C body coordinates
 ** Program variables:
 * A : 3x3 transformation matrix from S/C body coordinates. to 1-2-3 frame
 * A1 : area of central subsegment of unit S/C centered sphere,
 * defined as one unit area equal to the size of DA
 * BLOCKED : test flag. If the Lth CSS blockage blots out the
 * albedo light from the Kth infinitesimal area,
 * BLOCKED = 0.0 (zeros the DI from that area). If not
 * blocked, BLOCKED = 1.0
 * CGAMMA and SGAMMA : cosine and sine, respectively, of the angle
 * from axis of FOV (Field of View) cone to normal
 * of Kth dAe (deg)
 * CSSIMAX : maximum possible CSS output
 * D : distance from Kth dAe to the spacecraft's position
 * DI : CSS output current due to albedo from Kth dAe
 * HALFTHETA : equals THETA/2
 * HO : unit vector in direction of orbit normal; unit vector for axis 2
 * I : refers to lth concentric circle out of N concentric circles
 * J : refers to Jth radial area in lth annulus
 * K : refers to Kth unit area out of NA equal areas
 * KAPPA : reflective constant for Kth dA
 * L : refers to the Lth CSS out of P CSS
 * LIT : flag, indicates if Kth dA is in light or darkness
 * M : number of radial divisions in lth circle
 * NDOTU : CSS head unit vector dotted with Kth S/C to ground unit
 * vector to determine if light from Kth dAe is visible
 * to the spacecraft
 * NORM : outward pointing normal vector for Kth dAe
 * ONE : HO x RB; unit vector along axis 1
 * RI : distance from the spacecraft FOV cone's axis to area centers for lth
 * annulus
 * S123 : sun unit vector in 1-2-3 frame coordinates.
 * THETA : size of radial divisions in lth circle (deg)
 * U : unit vector from S/C to 'center' of Kth dAe (location of Kth
 * dA's NORM) in 1-2-3 coordinates
 * U123 : temporary holding vector for each U
 * USC : U transformed from 1-2-3 coordinates frame to S/C body coord.
 *** OUTPUT:
 * ALBICSS : final CSS currents due to albedo

IMPLICIT REAL*8 (A-H,O-Z)
 INTEGER I,J,K,L,M,N,NA,NATH,P,KL
 REAL*8 T,RE,ALTP,ALPHAMAX,PI,RMAG,HALFTHETA,AMAX,CSSIMAX
 REAL*8 NDOTU,CSSLM,THETA,RI,CGAMMA,SGAMMA,ALB,A1
 REAL*8 UDOTBLOCK,BLOCKAGE,BLOCKED
 REAL*8 DI,KAPPA,SDOTN,SB(3),S123(3),A(3,3),VB(3),RB(3),HO(3)

```

REAL*8 ONE(3),U123(3),USC(3),BLOCK(8,4)
REAL*8 U(1000,3),NORM(1000,3),D(1000),NCSS(3,10),ALBICSS(10)
REAL*8 LIT(1000),INALBED
COMMON/ALBFLAG/INALBED

```

```

*   begin initialization loop:
  IF (T .EQ. 0.0) THEN
*   set constants:
    BLOCKAGE = 1.0
    N = 6
    NA = (2*N-1)**2
    P = 8
    RE = 6378.0D0
    CSSLM = DCOSD(80.0D0)
    ALTP = 350.0
    RMAG = RE + ALTP
    PI = DACOS(-1.0D0)
    ALPHAMAX = DASIND(RE/RMAG)
    AMAX = 2*PI*(1-DCOSD(ALPHAMAX))
    A1 = AMAX/NA
    ALB = 0.30D0
    CSSIMAX = 1.0D0
    KAPPA = ALB*CSSIMAX*A1/PI

*   calculate unit vectors from spacecraft to Kth area normal
  K = 1
  DO I=1,N
    M = 8*(I-1)
    IF (M .EQ. 0.0) THEN
      U(K,1) = 0.0D0
      U(K,2) = 0.0D0
      U(K,3) = -1.0D0
      K = K+1
    ELSE
      THETA = 360.0D0/DBLE(M)
      HALFTHETA = 0.5D0*THETA
      RI = SQRT(DBLE(4*(I-1)*(I-1)+1))
      CGAMMA = (NA - RI**2 + RI**2*(DCOSD(ALPHAMAX)))/NA
      SGAMMA = SQRT(1-CGAMMA**2)
      DO J=1,M
        U(K,1) = SGAMMA*DCOSD(THETA*J - HALFTHETA)
        U(K,2) = SGAMMA*DSIND(THETA*J - HALFTHETA)
        U(K,3) = -CGAMMA
        K = K+1
      END DO
    ENDIF
  END DO

```

```

*   calculate area normals
    C = RMAG**2 - RE**2
    DO K = 1,NA
      B = RMAG*U(K,3)
      D(K) = -B-SQRT(B**2-C)
      NORM(K,1) = D(K)*U(K,1)/RE
      NORM(K,2) = D(K)*U(K,2)/RE
      NORM(K,3) = (RMAG+D(K)*U(K,3))/RE
    ENDDO

    ENDIF
*   end initialization loop

*   AT ALL TIMES:
*   FIND matrix A to go from s/c body to orbital frame 1-2-3
    CALL CROSSU(HO, RB, VB)
    CALL CROSSU(ONE, HO, RB)
    DO KL=1,3
      A(1, KL) = ONE(KL)
      A(2, KL) = HO(KL)
      A(3, KL) = RB(KL)
    ENDDO

*   transform sun unit vector from s/c body to 1-2-3
    CALL TO123(S123, A, SB)

*   Find current in Lth CSS due to albedo of the lit areas:
    DO L = 1, P
      ALBICSS(L) = 0.0D0
      DO K=1, NA
c       determine if Kth area is lit:
        SDOTN =
$ (S123(1)*NORM(K,1)+S123(2)*NORM(K,2)+S123(3)*NORM(K,3))
        IF (SDOTN .GT. 0.0D0) THEN
c         determine is the Kth dA is the Lth CSS's FOV:
          DO KL=1,3
            U123(KL) = U(K, KL)
          ENDDO
          CALL TOSC(USC, A, U123)

          NDOTU=(NCSS(1,L)*USC(1)+NCSS(2,L)*USC(2)+NCSS(3,L)*USC(3))
          IF (NDOTU .GT. CSSLM) THEN
c           If there is CSS blockage, determine if it blocks
c           the albedo from Kth area:
            BLOCKED = 1.0D0
            IF (BLOCKAGE .EQ. 1.0) THEN
              UDOTBLOCK=BLOCK(L,1)*USC(1)+BLOCK(L,2)*USC(2)
$              +BLOCK(L,3)*USC(3)

```

```

        IF (UDOTBLOCK .GT. DCOSD(BLOCK(L,4))) BLOCKED = 0.0D0
        ENDIF
        DI = KAPPA*NDOTU*SDOTN*BLOCKED
        ALBICSS(L) = ALBICSS(L) + DI
        ENDIF
    ENDIF
ENDDO
ENDDO

RETURN
END

```

```

* SUBROUTINE CROSSU(C,A,B)
  C = AxB; normalize C and return a unit vector
  INTEGER I
  REAL*8 C(3),A(3),B(3)
  C(1) = A(2)*B(3)-A(3)*B(2)
  C(2) = A(3)*B(1)-A(1)*B(3)
  C(3) = A(1)*B(2)-A(2)*B(1)
  CMAG = SQRT(C(1)**2+C(2)**2+C(3)**2)
  DO I=1,3
    C(I) = C(I)/CMAG
  ENDDO
  RETURN
  END

```

```

* SUBROUTINE TO123(V123,A,VSC)
  transforms a vector from S/C body to 1-2-3 orbital frame
  INTEGER I,J,K
  REAL*8 V123(3),A(3,3),VSC(3)
  DO J=1,3
    DO I=1,3
      V123(J) = V123(J) + A(J,I)*VSC(I)
    ENDDO
  ENDDO
  VMAG = SQRT(V123(1)**2+V123(2)**2+V123(3)**2)
  DO K = 1,3
    V123(K) = V123(K)/VMAG
  ENDDO
  RETURN
  END

```

```

* SUBROUTINE TOSC(VSC,A,V123)
  transforms a vectors from orbital frame 1-2-3 to S/C body
  INTEGER I,J,K
  REAL*8 V123(3),A(3,3),VSC(3)
  DO I=1,3
    DO J=1,3

```

```
VSC(I) = VSC(I) + A(J,I)*V123(J)
ENDDO
ENDDO
VMAG = SQRT(VSC(1)**2+VSC(2)**2+VSC(3)**2)
DO K=1,3
  VSC(K) = VSC(K)/VMAG
ENDDO
RETURN
END
```

Appendix II

Summary of Reflectance Data for Earth Surface Features and Clouds

Reflecting	Magnitude and Other	Angular Distribution of	Total Reflectance
Soils and Rocks	Increases to 1 μm Decreases above 2 μm .	Backscattering and forward scattering. Sand has large forward scattering. Loam has small forward scattering.	5 to 54 percent. Moisture decreases reflectance by 5 to 20 percent. Smooth surfaces have higher reflectance. Diurnal variation. Maximum reflectance for small Sun angles.
Vegetation	Small below 0.5 mm. A small maximum bump at 0.5 to 0.55 mm. Chlorophyll absorption at 0.68 μm . Sharp increase at 0.7 mm. Decreases above 2 μm . Depends on growing season.	Backscattering. Small forward scattering.	5 to 25 percent. Diurnal effects. Maximum reflectance for small angles. Marked annual variation.
Water Basins	Maximum at 0.5 to 0.7 μm . Depends on turbidity and waves.	Large back and forward scattering.	50 to 20 percent. Diurnal variation. Maximum for small Sun angles depends on turbidity and waves.
Snow and Ice	Decreases slightly with increasing wavelength. Large variability depends on purity, wetness, and physical condition.	Diffuse component plus mirror component. Mirror component increases with increasing angle of incidence.	Variable 25 to 80 percent. 84 percent in Antarctic. 74 percent in Ross Sea ice. 30 to 40 percent in White Sea ice.
Clouds	Constant from 0.2 mm to about 0.8 μm . Decreases with wavelength above 0.8 μm , showing water vapor absorption bands.	Pronounced forward scattering with small backscattering. Minimum for scattering angles of 90° to 120°. Fogbow* for scattering angle of 143°.	10 to 80 percent Varies with cloud type, cloud thickness, and type of underlying surface.

*A nebulous arc or circle of white or yellowish light sometimes seen in a fogbank.

SOURCE: "Earth Albedo and Emitted Radiation"

REPORT DOCUMENTATION PAGE

Form Approved
OMB No. 0704-0188

Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project (0704-0188), Washington, DC 20503.

1. AGENCY USE ONLY (Leave blank)		2. REPORT DATE January 1994	3. REPORT TYPE AND DATES COVERED Reference Publication	
4. TITLE AND SUBTITLE <i>An Earth Albedo Model - A Mathematical Model for the Radiant Energy Input to an Orbiting Spacecraft Due to the Diffuse Reflectance of Solar Radiation From the Earth Below</i>			5. FUNDING NUMBERS 710	
6. AUTHOR(S) Thomas W. Flatley and Wendy A. Moore				
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS (ES) Goddard Space Flight Center Greenbelt, Maryland 20771			8. PERFORMING ORGANIZATION REPORT NUMBER 94B00034	
9. SPONSORING / MONITORING AGENCY NAME(S) AND ADDRESS (ES) National Aeronautics and Space Administration Washington, DC 20546-0001			10. SPONSORING / MONITORING AGENCY REPORT NUMBER NASA TM-104596	
11. SUPPLEMENTARY NOTES Flatley and Moore: Goddard Space Flight Center, Greenbelt, Maryland				
12a. DISTRIBUTION / AVAILABILITY STATEMENT Unclassified - Unlimited Subject Category 12			12b. DISTRIBUTION CODE	
13. ABSTRACT (Maximum 200 words) Past missions have shown that the Earth's albedo can have a significant effect on the sun sensors used for spacecraft attitude control information. In response to this concern, an algorithm was developed to simulate this phenomenon, consisting of two parts, the physical model of albedo and its effect on the sun sensors. This paper contains the theoretical development of this model, practical operational notes, and its implementation in a FORTRAN subroutine.				
14. SUBJECT TERMS Albedo; Reflectance; Sun Sensors; Solar Flux; Solar Radiation; Reflection; Spacecraft Sensors; Light; Course Sun Sensors			15. NUMBER OF PAGES 44	
			16. PRICE CODE	
17. SECURITY CLASSIFICATION OF REPORT Unclassified	18. SECURITY CLASSIFICATION OF THIS PAGE Unclassified	19. SECURITY CLASSIFICATION OF ABSTRACT Unclassified	20. LIMITATION OF ABSTRACT UL	

National Aeronautics and
Space Administration
Code JTT
Washington, D.C.
20546-0001

SPECIAL FOURTH-CLASS RATE
POSTAGE & FEES PAID
NASA
PERMIT No. G27

Official Business
Penalty for Private Use, \$300



POSTMASTER: If Undeliverable (Section 158,
Postal Manual) Do Not Return
